
Abstract 
 

The commonest non-SAS tool for data presentation, and working with data, is Microsoft Excel. Other people may want your SAS report 

formatted as a spreadsheet. Dynamic Data Exchange (DDE) can empower and augment your use of SAS with Excel. You can develop SAS 

programs to load worksheets, and format them without point-and-click. Almost anything that you can do directly in Excel can also be done 

using DDE commands from your SAS program. You can access Excel data, even to the level of a specific cell, for writing, reading, or 

formatting data. To eliminate worry about getting code details right every time, you can simplify and routinize your use of DDE with 

macros. The presentation shows two different ways to simultaneously deliver data at summary and detail levels in one Excel workbook. 

Nowhere-else-available technical tidbits are included. E.g., custom indenting of cell contents is unsupported by DDE, but you will learn 

how to accomplish it. Colour-coding the display of critical data elements, to signal value significance, is easily and often implemented for 

ODS tables—here a not-obvious solution for Excel is provided to you. The intended audience is all levels of SAS users. 

 

Introduction 
 

Why would you want to run Excel from SAS? You can create a hands-off production job to be automatically scheduled by the computer to 

access and analyze the data, load the spreadsheet, format it, and attach the report to a SAS-dispatched email message. In any case, if you 

find yourself preparing the same Excel report over and over, why not automate it? 

 

This paper and the associated slide presentation are an introduction to a self-study collection of sixty macros and over twenty sample 

programs (with substantial comments), and supporting files, that are available for SAS with Excel application development, via email. 

To find out more about the Excel commands used here to work with DDE, or to find other commands or command options not used here, 

you must consult the documentation in Macrofun.hlp. To get this information, you must install Macrofun.hlp by initially downloading 

Macrofun.exe from the Microsoft Download Center, and then running the .exe file. If you are not experienced using the Download Center, 

you can find the right part of it by using google.com to search for “macrofun.exe”, or instead you could try this URL: 

http://support.microsoft.com:80/support/kb/articles/Q128/1/85.ASP.  

 

NOTES: To keep the paper compact, screen images have usually been clipped to show only the essential parts of the displayed result. 

Also, rather than try to explain, incompletely, the options and syntax rules for DDE commands used, I recommend your looking into 

Macrofun.hlp for the complete story. 

 

DDE: Dynamic Data Exchange 
 

Client-Server Architecture 
 

Your SAS session is a client. It opens Excel, which acts as a server. Your SAS program: 

 

1. Starts/Executes Excel 

2. Establishes a channel to talk to Excel 

3. Opens a new/existing XLS file (workbook) 

4. Sends commands to Excel 

5. Saves, or Saves As (new/diff), XLS file 

6. Exits/Stops Excel 

 

What Can You Do with DDE? 
 

Anything that you could do in Excel with your keyboard and mouse!!! 

 

Here I cover only a subset of the input/output and formatting possibilities. When DDE can not do it (rare), you can pre-record an Excel 

macro and run it with DDE. DDE is more flexible than simple Excel macros. With DDE, you programmatically do anything that you can 

do with a mouse and keystrokes in an Excel window. In fact, if you configure your client PC appropriately, you can watch Excel working 

as a robot under the direction of your program. The same visual effects that appear on your screen when your hands are doing the work 

appear when your program is in control. 

 

Some of the Cell Formatting Options 
 

• Font controls available includes: font, style, size, colour, underline. 

• Cell background can be filled with colour. 

• Cell sizing is possible, and column or row AutoFit and column or row Hide/Unhide. 



• Cell alignment, cell merge, text wrap, etc. can be done. 

• Borders can be created, e.g., to section the worksheet. 

• Cell content can be indented, using two tricks shown in sample programs, using macros. 

 

Some of the Worksheet Controls 
 

• Select worksheet, cells, rows, columns 

• Freeze/Unfreeze panes 

• Turn AutoFilter On/Off 

• Generate subtotals in a worksheet 

• Control/Change levels for Excel subtotals 

(or use multiple sheets for levels of totals) 

• Activate a specific worksheet in a workbook 

• Create additional worksheets in a workbook 

• Delete unused worksheets in a workbook 

 

Some of the Other Manipulation Possibilities 
 

• Insert/Delete columns or rows 

• Find, Find and Replace cell contents 

• Copy cell, Move cells 

• Use formulas to create new column(s) 

 

Simple Example of Using Two Miscellaneous DDE Functions 
  

You can use Zoom, to shrink (or enlarge) a worksheet, and Message, to access the text box in the lower left corner of the window.

 

 

For the Original Worksheet: 

 
 

Running this code: 

 
data _null_; 
file DDEcmds; * How this fileref is 
     established will be explained later *; 
put '[zoom(80)]'; 
put '[message(true, 
"This Sheet reduced to 80% of normal 
size")]'; 
run; 

 

Yields this result: 

 
 

The original cell/worksheet size and the default worksheet 

message (“Ready”) are restored with this code: 

 
data _null_; 
file DDEcmds; 
put '[zoom(100)]'; 
put '[message(false)]'; 
run;



The Mechanics of DDE and the Basic SAS-with-Excel Interaction 
 

Channel for DDE Commands 
 

DDE requires use of special SAS filename statements: 

 
filename YourFileRef dde "excel|system"; 
 

YourFileRef is arbitrary 1-to-8 characters. I like DDEcmds. 

 

How I Handle the DDE Fileref 
 

As initialization code in my sample programs and/or macros you will see 

 
filename &ddefileref dde "excel|system"; 
 

Earlier in the initialization processing, there is always the statement 

 
%let ddefileref = DDEcmds; 
 

Why &ddefileref? Avoid loss of generality. The “DDEcmds” fileref is not mandatory. Choose your own fileref.  

 

The DDE Triplet for Excel 
 

Identification of the current selection of worksheet and a rectangle of its cells for read or write is accomplished with: 

 
EXCEL|YourSheet!RpppppCqqq:RsssssCttt 
 

ppppp is 1-to-5-digit start/top row number 

qqq is 1-to-3-digit start/left column number 

sssss is end/bottom row number 

ttt is end/right column number 

 

Row/column numbers with leading zeros are permissible. 

 

You can simplify identification of the certain selections of cells as follows: 
 

RpppppCqqq    - single cell 

Rppppp:Rsssss - range of rows 

Rppppp        - single row 

Cqqq:Cttt     - range of columns 

Cqqq          - single column 

 

Coding the Cell Selection 
 

You are not required to permanently hard code it. You can use a macro variable, and Software Intelligence (SI) in your program can 

dynamically determine the number of rows or ending row required, etc. 

 

What is “Software Intelligence”? This is a term coined by LeRB in the early 1990’s. SI programs can be more than merely data-adaptive. 

Full explanation and illustrations of SI Application Development are available in other papers by the author. 

 

The paper and presentation show three ways to code the DDE triplets, etc.: (a) hard-coding; (b) the author’s SAS macros; and SI generation 

based on data characteristics. 

 

Excel DDE Commands 
 

DDE commands were originally developed by Microsoft for earlier editions(s) of Excel. I have used them with various versions of 

Windows and Excel. They almost all work for me, but I can offer no guarantees. 

 

The commands are documented in Macrofun.hlp, which you can download from Microsoft as mentioned above. “Macrofun” is an 

abbreviation for “Macro Functions”, not “Macro Fun”, but these macros functions are fun to use. They do not require use of Excel macros. 

 

Please be aware that not every DDE command works (e.g., rename worksheets, create subtotals, . . .), and some command options may not 

work. Nevertheless, what does work is an enormous tool set! 



true / false Values for DDE Commands 
 

These are used to turn options On / Off in DDE command parameter assignments. But be aware that for some commands you are required 

to use 1 / 0 instead. 

 

The Basic DDE Functions 
 

1. Start Excel: provides a default empty workbook, with three worksheets 

2. Open an existing workbook 

3. Save As of the current workbook with a new name 

4. Exit Excel 

 

Starting Excel from SAS 
 

For other versions of MS Office or Excel than that used for my development, the folder sequence in code below may differ: 

 
x '"C:\Program Files\Microsoft Office\Office\EXCEL.exe"'; 
data _null_; 
z=sleep(3); /* wait 3 seconds for Excel to start */ 
run; 

 

The above program opens a new workbook, with three empty worksheets: Sheet1, Sheet2,and Sheet3. 

 

Open Existing Workbook from SAS 
 
data _null_; /* talk to DDE, no output data */ 
file DDEcmds; 
put '[open("c:\YourFolder\YourWorkbook.xls")]'; 
x=sleep(3); /* wait 3 seconds for it to open */ 
run; 

 

Finished Workbook Save As 
 
data _null_; 
file DDEcmds; 
put '[error(false)]'; 
put '[save.as ("c:\OtherFolder\DifferentFileName.xls")]'; 
x=sleep(1); 

run; 

 

Exiting Excel from SAS 
 
data _null_; 
file DDEcmds; 
put '[error(false)]'; 
put '[quit()]';  /* empty parenthesis */ 
run; 

 

The error(false) command tells Excel that you do not want any prompts to confirm intention. 

 

Overview of Custom SAS Macro Solutions (You can skip this, permanently, or come back later) 
 

Macros for the Same Scenario As Above 
 

You may not want to have to code so many lines of, frankly, unintuitive code. Available from the author are macros that could be used, to 

accomplish the same results, as follows: 

 
%XLStart; 
%XLDDEcmdsFileName; 
%XLOpen(WorkBook=c:\YourFolder\YourWorkbook.xls); 
%XLSaveAs(WorkBook=c:\OtherFolder\DifferentFileName.xls); 
%XLExit; 



Macro for Starting Excel 
 
%macro XLStart(ExcelPgm= 
C:\Program Files\Microsoft Office\Office\EXCEL.exe,SecondsToWaitForStart=3); 
options noxwait noxsync; 
x %unquote(%str(%'"&ExcelPgm"%'));  
data _null_; 
z=sleep(&SecondsToWaitForStart);run; 
%mend  XLStart; 

 

Macro for Opening an Existing Workbook 
 
%macro XLOpen(WorkBook=,SecondsToWaitForOpen=3); 
data _null_; 
file DDEcmds; 
put %unquote(%str(%'[open("&Workbook")]%')); 
z=sleep(&SecondsToWaitForOpen); 
run; 
%mend  XLOpen; 

 

Macro for Saving a Workbook 
 
%macro XLSaveAs(WorkBook=); 
data _null_; 
file DDEcmds; 
put '[error(false)]'; 
put %unquote(%str(%'[save.as("&WorkBook")]%')); 
x=sleep(1); 
run; 
%mend XLSaveAs; 

 

When Argument(s) Is (Are) Macro/Symbolic Variable(s) 
 

The value of argK is put in the symbol table by: 

 
%YourMacro(argK=FormattedValue); 
or 
%let argK = FormattedValue; 
or 
call symput('argK',put(value,format.)); 
 

argK is retrieved from the symbol table with: 

 
put %unquote(%str(%'[command(&argK)]%')); 
 

OPTIONS MPRINT shows the resolved and executed statement to be: 

 
put '[command(FormattedValue)]'; 
 

My SAS-with-Excel Libraries of Programs and Macros 
 

Available are: 

 

• 26 sample programs 

• 5 empty, but specially preformatted, spreadsheets to use in sample programs 

• Text files of colour definition information 

• 60 macros for your use 

• Abundant example code to invoke the macros, which is most educationally run with “OPTIONS MPRINT;” 

 

Running example programs with “OPTIONS MPRINT;” shows generated SAS code in the SAS log. This enables you to understand what 

any macro does, and to understand what the DDE command content requirements are. 

 

If you wish to avoid use of a macro, you can copy generated code from the SAS log, strip off the MPRINT prefix at the left margin, and 

adapt that code to other uses without the macro. 



Please note these caveats. All macros & programs were tested, but may be incompatible with your software. They are NOT “industrial-

strength”. I.e., they assume that you will use valid macro parameter assignments. If you add the needed editing, I will use it, and publish it 

with attribution to you. You should test the macros before assuming that their results are correct. I can only offer them “As Is”. 

 

To Use a Macro Library 
 

Save the macro in a Windows folder (e.g., c:\MyMacLib). The SAS-Instituted supplied macros are referred to with fileref “SASAUTOS”, 

which is also the label of a SAS system option. You must identify the location of your macros: 

 
filename MyMacros 'c:\MyMacLib'; 
 

and then tell SAS to search your macro library (if first, with this parameter sequence): 

 
options sasautos=(MyMacros sasautos); 
 

A Tour of DDE Applications 
 

We will start with a workbook which is loaded with data from SASHELP.CLASS and uses the default Excel formatting. 

 

 
 

Font Formatting – Step 1 
 

Activate worksheet (if not already there). Not really necessary—this workbook has only one worksheet. Select the cells to be formatted 

 
data _null_; 
file DDEcmds; 
put '[workbook.activate("CLASS")]'; 
put [select("R5C3:R7C5")]'; 
run; 

 

Cells to be formatted Selected: 

 



Font Formatting – Step 2 
 

Let’s use: 

• font Verdana 

• style Bold 

• size 16 point 

• (Excel) colour 3 

 

Here’s how (those commas are very important, and represent unused but available options): 

 
data _null_; 
file DDEcmds; 
put '[font.properties("Verdana","Bold",16,,,,,,,3)]'; 
run; 
 

The font is changed, but is obscured by the cell selection box and shading: 

 
 

Font Formatting – Step 3 for Demo 
 

After the block of cells has been formatted, the necessary predecessor select("R5C3:R7C5")command leaves them highlighted (just 

as a mouse select would). But I want to see the new font colour.  So, move the focus with: 

 
data _null_; 
file DDEcmds; 
put '[select("R1C1")]'; 
run; 

 

“I can see clearly now.” 

 
 

Font Formatting – All In One Step 
 
data _null_; 
File DDEcmds; 
put '[workbook.activate("Class")]'; 
put '[select("R5C3:R7C5")]'; 
put '[font.properties("Verdana","Bold Italic",16,,,,,,,3)]'; 
put '[select("R1C1")]'; run;



DDE Command Format 
 

In a command of the form 

 
put '[CommandName(arg1,arg2,...,argN)]'; 
 

the CommandName may be multi-word, dot-separated. 

 

For an example with several (in this case, four) command arguments, there are three ways to indicate omission of one argument— 

 

• at start of list: ,arg2,arg3,arg4 

• in middle of list: arg1,,arg3,arg4 

• at end of list: arg1,arg2,arg3   (i.e., just omit) 

 

Font Formatting with Macros 
 
%XLSelect(WorkSheet=Class,Row=5,Column=3,ToRow=7,ToColumn=5) 
%XLFont(FontName=Verdana,FontStyle=Bold, FontSizePts=16,FontColorNumber0To56=3); 
%XLFocus; 
 

Colour the Cell Background (Using Macros) 
 
data _null_; 
file DDEcmds; 
* for demo, select last cell in sheet *; 
put '[select.special(11)]';  
run; 
%XLBackgroundColor(Color=3); /* 3 = Red */ 
* now move the Select to see cell colour *; 

%XLFocus; 

 

Colour the Cell Background (Hard Coded) 
 
data _null_; 
file DDEcmds; 
put '[select.special(11)]'; 
put '[patterns(1,,3,true)]'; 
put '[select("R1C1")]'; 
run; 
 

 



Remove the Background Colour 
 
data _null_; 
file DDEcmds; 
put '[undo()]'; 
run; 

 

Colour Undone, Cell Still Selected: 

 
 

Prep to Colour-Code the Height Column 
 

We need to be able to read all the data cells from Column 4. 

 
filename Column DDE "EXCEL|Class!R2C4:R65536C4" notab; 
 

Row 1 contains column headings, but we, a priori, do not know the number of actively used data rows. So, with the filename statement, 

we prepared SAS to read Row 2 through the Excel maximum Row 65536.  Now we can read the data from the Height column and compute 

the average.  

 
data AllHeights; 
infile Column; 
input height; 
run; 
 
proc means data=AllHeights noprint mean; 
var height; 
output out=AverageHeight mean=AvgHgt; 
run; 

 

We need some criterion for colour-coding, and will arbitrarily designate taller than average as “Bad”, and shorter than average as “Good”. 

 
data _null_; 
set AllHeights end=LastCell; 
if _N_ eq 1 then set AverageHeight; 
if height le AvgHgt then ColorIt = 'Good'; 
else ColorIt = 'Bad'; 
call symput('ColorIt' || trim(left(_N_)),trim(left(ColorIt))); 
run; 

 

What Did the Prep Do? 
 

If you submit 

 
%put _user_;  

 

in the SAS log, you will see this listing of symbol table content: 



GLOBAL COLORIT1 Bad 
GLOBAL COLORIT2 Good 
GLOBAL COLORIT3 Bad 
  .  .  . 
GLOBAL COLORIT18 Good 
GLOBAL COLORIT19 Bad 
GLOBAL HOWMANY           19 

 

For brevity, we will use the author’s macro to do the colour-coding. (In principle, a macro could also be developed to do the colour-coding 

prep step as well. 

 
%XLColorCodeCellsInColumn( 
Column=4,StartRow=2, 
ColorCodeSymVarPrefix=ColorIt, 
ColorCodeCount=&HowMany, 
GoodXLBackgroundColorNumber=32, /* Blue  */ 
GoodXLForegroundColorNumber=2,  /* White */ 
BadXLBackgroundColorNumber=3,   /* Red   */ 
BadXLForegroundColorNumber=1);  /* Black */ 
* now move Select, to see the new colour *; 
%XLFocus; 

 

The particular combinations of foreground (text) colour and background (fill) colour are selected to maximize readability of the cell 

content. The screen print (clipped and magnified) does not demonstrate it well, but White on Blue is OK. Black on Blue would be difficult 

to read. 

 

Screen Display of White Text OK, Screen Print Not As Good: 
 

 



Using Formulas in Excel 
 

Normally, one could create all the data in SAS pre-spreadsheet-load processing. Suppose, however, we want columns derived from Excel 

subtotals. For demo only, we will work with simple spreadsheet content, without subtotals. With subtotals present, the use of Excel 

formulas works the same. 

 

Insert a New Column Left of Column 4 
 
data _null_; 
file DDEcmds; 
put '[select("C4")]'; 
put '[insert(4)]'; /* 4 specifies type of INSERT */ 
run; 

 

 
 

Excel Format the Cells in Column 4 Like SAS Format 4.1 
 
 /* no need to re-select the column, because the focus is still there */ 
data _null_; 
file DDEcmds; 
put '[Format.Number("#0.0")]'; 
run; 

 

Insert a Heading for the New Column 4 
 
filename CELL1 DDE "EXCEL|Class!R1C4"; 
data _null_; 
file CELL1; 
put 'Lbs./Yr.'; 
run; 

  

 
 

Prepare for a Division Operation 
 

Column 3 is to be the divisor. For this demo, set some of its cells to null (missing) values, and other cells to zero values. 

 
data _null_; 
file DDEcmds; 
put '[select("C3")]'; 
put '[formula.replace("16","",1,,false,false)]'; 
put '[formula.replace("12","0",1,,false,false)]'; 
put '[select("R1C3")]'; /* remove highlight */ 
run; 



 
 

Divide Weight By Age 
 
data _null_; 
file DDEcmds; 
put '[select("R2C4")]'; 
put '[formula.fill("=RC[+2]/RC[-1]","R2C4:R20C4")]'; 
run; 

 

 
 

Get Rid of the Error Message #DIV/0! 
 
data _null_; 
file DDEcmds; 
put '[select.special(3,16)]'; 
put '[clear]'; 
run; 

 

Cleared Cells Remain Selected: 

 



Replace Nulls with “N/A” 
 
data _null_; 
file DDEcmds; 
put '[formula.replace("","N/A",1,,false,false)]'; 
put '[select("R1C4")]'; /* remove the highlight */ 
run; 

 

Almost Finished: 

 
 

Make the Headings Bold, and AutoFit the Columns 
 
data _null_; 
file DDEcmds; 
put '[select("R1")]'; 
put '[font.properties("","Bold")]'; 
                   /* "", = font name unchanged */    
put '[select("C1:C6")]'; 
put '[column.width(,,,3)]'; /* commas required */ 
                   /* 3 = AutoFit */  
put '[select("R1C1")]'; /* remove the selection highlight */ 
run; 

 

Better, But Still Not Finished: 

 



Align Columns 2 and 3 
 
data _null_; 
file DDEcmds; 
put '[select("C2:C3")]'; /* Sex & Age Columns */ 
put '[alignment(3,false,3,0)]'; /* 3 = Center */ 
put '[select("C4")]'; /* Lbs. / Yr. Column */ 
put '[alignment(4, false,3,0)]'; /* 4 = Right */ 
put '[select("R1C1")]'; 
run; 

 

Need To Fix the Format of Height and Weight: 

 
 

Excel Format the Height and Weight Data Like SAS Format 4.1 
 
data _null_; 
file DDEcmds; 
put '[select("C5:C6")]'; 
put '[format.number("#0.0")]'; 
put '[select("R1C1")]'; * remove the column selection highlight *; 
run; 

 

 



Customized Excel Worksheet with Subtotals 
 

We start with a worksheet pre-loaded via export of SASHELP.CLASS. 

 

 
 

Now we sort the data with: 

 
data _null_; 
file DDEcmds; 
put '[select("R1C1:R65536C256")]; /* all cells */ 
put '[sort(1,"Sex",1,"Age",1,"Name",1,1,1,0)]'; 
put '[select("R2C1")]'; run; 
 
/* 1 = sort by rows */ 
/* ,"Sex",1,"Age",1,"Name",1 = sort on Sex, then Age, then Name, all ascending */ 
/* 1,1,0 = Normal sort, Headers present, Not case sensitive */ 

 

The worksheet has been sorted by Name within Age within Sex: 

 
 

After closing the data workbook, open a workbook of only Excel macros: 

  
data _null_; 
file DDEcmds; 
put '[close(true)]'; 
x=sleep(1); 
put '[open("C:\Folder\ExcelMacros.xls")]'; 
x=sleep(3); run; 



If pasted in here, the screen image of the workbook of only Excel macros would simply show an empty Sheet1. 

 

Using Excel “hands-on” to get to it, for illustration only, here is a pasted-in, clipped, magnified image of the Excel macro code: 

 
 

Now open the data workbook, and run the sasHelpClass_SubtotalsByAgeWithinSex Excel macro: 

  
data _null_; 
file DDEcmds; 
put '[open("C:\Folder\TempClass.xls")]'; 
x=sleep(3); 
put '[run("ExcelMacros.xls!sasHelpClass_SubtotalsByAgeWithinSex",false)]'; run; 

 

Excel-Generated Subtotals with Level Buttons, But Grand Totals Twice: 

 



Reformat & Save 
 
data _null_; 
file DDEcmds; 
put '[select("R1")]'; 
put '[font.properties("","Bold")]'; 
put '[select("C1:C5")]'; 
put '[column.width(,,,3)]'; 
put '[error(false)]'; 
put '[save.as("C:\Folder\DataWithSubtotals.xls")]'; 
x=sleep(1); 
run; 

 

Remove Extra Grand Total 

 
Find the last row, and put that row number in the symbol table: 

 
filename Column DDE "EXCEL|CLASS!R1C1:R65536C1" notab; 
data _null_; 
length FirstCharInColumnCell $ 1; 
infile Column truncover end=LastObs; 
input FirstCharInColumnCell; 
if LastObs; 
call symput("LastRow",trim(left(_N_))); 
run; 

 

Find the second from last row, and delete it: 

 
data _null_; 
LastRow = &LastRow; 
SecondLastGrandTotal = LastRow - 1; 
call symput('SecondLastGrandTotal',trim(left(SecondLastGrandTotal))); 
run; 
 
data _null_; 
file DDEcmds; 
put '[select("R' "&SecondLastGrandTotal" '")]'; 
/* resolves to: put '[select("R34")]'; */ 
put '[edit.delete(3)]'; run; 

 

Only One Grand Total: 

 
 

Pseudo-Indent Cell Content  
 
data _null_; 
file DDEcmds; 
put '[select("R2C2:R15C2")]'; 
put   '[formula.replace("F","___Female",1,,false,false)]'; 
  /* change colour of three _ to White 
     to match cell background colour: */ 



put '[font.properties(,,,,,,,,,2,,,1,3)]'; 
run;  /* 2: colour the text string White 
         1: start at character 1 
         3: for a length of 3 characters */ 

 

Pseudo-Indents Complete: 

 
 

 
 

ReLabel Sex Subtotals (Using Macros Available from the Author) 
 
%XLSelectColumns(Column=2); 
%XLReplaceCellContents(Find=F Total,Replace=All Females); 
%XLReplaceCellContents(Find=M Total,Replace=All Males); 
%XLColumnWidth(Column=2,ToColumn=2,Action=AutoFit); 
%XLFocus; /* Use %XLFocus to De-Select Column 2 */ 

 

New Labels for Sex Subtotals: 

 
 

 
 

 
 

Excel subtotals above are showing to Level 4. Let’s exercise the subtotals level display controls. 

 

Show Subtotals Only to Level 3 
 
data _null_; 
file DDEcmds; 
put '[show.levels(3,0)]'; 
run;



Excel Subtotals, Showing to Level 3: 

 
 

Excel Subtotals, Showing to Level 2: 

 
 

Excel Subtotals, Showing to Level 1: 

 

 

Another Way To Do Subtotals 

 

You can really “Have It Your Way” with this custom design and construction.  It relies a one-time preparation of multiple pre-formatted 

empty worksheets, which are loaded via DDE. In an analogue or extension of this application, you may have to use Software Intelligence 

for post-load formatting. The user moves from “level” to “level” via clicking on worksheet tabs, not Excel “level buttons”. Here we show 

only the results, not the code (which is available, on request, from the author). 

 

Overview WorkSheet: 

 
 

Summary WorkSheet: 

Detail + Summary WorkSheet: 

 

Detail-Only WorkSheet: 



Bibliography (Annotated) 
 

There is an introduction to DDE in the “SAS Companion for the Microsoft Windows Environment” (published by SAS Institute Inc., Cary, 

NC, USA), even in the Version 6 Edition. 

 

If your needs are very simple (i.e., just transferring data from SAS to Excel, or vice versa), you might find this survey of alternative 

methods, not just DDE, very useful— 

 

David Rucker, “Not All Fish Eat Worms: A SAS Programmer’s Guide to MS Excel and Other Fish Stories”, Proceedings of the Twenty-

Eighth Annual SAS Users Group International Conference, SAS Institute Inc. (Cary, NC, USA), 2003. 

 

For a more sophisticated way to start a SAS-with-Excel application, for which you might have need (I have never needed this method), 

see— 

 

Christopher A. Roper, “Intelligently Launching Microsoft Excel from SAS, using SCL Functions Ported to Base SAS”, Proceedings of the 

Twenty-Fifth Annual SAS Users Group International Conference, SAS Institute Inc. (Cary, NC, USA), 2000. 

 

For the work of the most frequent author/presenter on the topic of SAS with Excel via DDE, see— 

 

Koen Vyverman, “Using Dynamic Data Exchange to Pour SAS Data into Microsoft Excel”, Proceedings of the Eighteenth SAS Users 

European Users Group International Conference, 2000. 

 

Koen Vyverman, “Using Dynamic Data Exchange to Export Your SAS Data to MS Excel – Against All ODS, Part I”, Proceedings of the 

Twenty-Sixth Annual SAS Users Group International Conference, SAS Institute Inc. (Cary, NC, USA), 2001. 

 

Koen Vyverman, “Creating Custom Excel Workbooks from Base SAS with Dynamic Data Exchange”, Proceedings of the Twenty-Seventh 

Annual SAS Users Group International Conference, SAS Institute Inc. (Cary, NC, USA), 2002. 

 

It could be sufficient to read only the last of Vyverman’s papers. Besides providing an excellent introduction to SAS with Excel, he 

addresses some difficult problems, in which you might be interested. These include: (a) how to rename worksheets; (b) how to create Excel 

macros from SAS (rather than pre-record them in Excel for later use by SAS); and (c) how to send keystrokes to Excel—the solution of last 

resort.  

 

Vyverman does take up some of the same matters that I present, but I cover more of the Excel macro commands, and work on several other 

topics, without taking up the special problems mentioned above. 

